

第5章 回帰分析の諸問題（2）

第1節 分布ラグ

1. 分布ラグの例

表3-3のデータについて、分布ラグモデルを適用する。次のプログラムを入力し、実行してみよう。（ファイル名は **ex3-3lag.tsp** としておく）

```
freq a;
smpl 1983 2002;
read(file='h:ex3-3.xls');
?
? ブラウン型消費関数
?
olsq ce c yd ce(-1);
ste=@res/@s;
graph date ste;
?
? アーモン・ラグの変数変換
?
wyd1=yd+yd(-1)+yd(-2)+yd(-3);
wyd2=yd(-1)+2*yd(-2)+3*yd(-3);
wyd3=yd(-1)+4*yd(-2)+9*yd(-3);
?
? パラメータの推定
?
print ce wyd1 wyd2 wyd3;
olsq ce c wyd1 wyd2 wyd3;
?
? もとのパラメータへの変換
?
freq n;
smpl 1;
b1=@coef(2);
b2=@coef(2)+@coef(3)+@coef(4);
b3=@coef(2)+2*@coef(3)+4*@coef(4);
b4=@coef(2)+3*@coef(3)+9*@coef(4);
print b1 b2 b3 b4;
end;
```

第2節 ダミー変数

2. 一時ダミーの例

1985年から2002年までの日本の水稻の作付面積と収穫量のデータについて、一時ダミーを入れたモデルを考え、分析してみる。水稻の作付面積と収穫量のデータ(dummy1.xls)を、講義用HPからダウンロードし、保存しておくこと。

その上で、次のプログラムを入力し実行してみよう。(ファイル名は dummy1.tsp としておく)

```
freq a;
smpl 1985 2002;
read(file='h:dummy1.xls');
?
? 通常の単回帰
?
olsq y c x;
ste1=@res/@s;
graph date ste1;
?
? 一時ダミーを入れたモデル
?
load d;
8*0 1 9*0;
print y x d;
olsq y c x d;
ste2=@res/@s;
graph date ste2;
end;
```

3. 季節調整ダミーの例

表5-4のデータについて、季節調整ダミーを用いて分析をおこなってみる。TSPではプログラムにおいて季節ダミーを生成することができる。実質家計最終消費支出(CE)と実質家計可処分所得(YD)のみが、ex5-4.xlsにあるので、これをダウンロードして保存しておこう。そして次のプログラムを入力し、実行してみよう。(ファイル名は ex5-4.tsp としておく)

```
freq q;
smpl 1996:1 2003:1;
read(file='h:ex5-4.xls');
?
? 散布図の表示とダミー変数の準備
?
graph yd ce;
dummy(exclude,prefix=d);
print ce yd d1-d3;
?
? 季節調整ダミーを入れた分析
?
olsq ce c yd d1-d3;
ste=@res/@s;
graph yd ste;
end;
```

4. 構造変化の検定と係数ダミーの例

表 3-3 のデータを用いて、CE を YD に対して単回帰をおこなうことを考える。この期間は 1980 年代のバブル期とバブル崩壊後の低成長期を含んでおり、消費構造の変化が考えられる。そこで、バブル崩壊前後（1991 年前と 1992 年以降）で、構造変化があったかを検定し、構造変化があったならば、定数項ダミーと係数ダミーを入れて分析をおこなってみる。

次のプログラムを入力し、実行してみよう。（ファイル名は **dummy2.tsp** としておく）

```
freq a;
smpl 1983 2002;
read(file='h:ex3-3.xls');
?
? チャウ検定
?
graph ce yd;
regopt(chowdate=1992) chow;
olsq ce c yd;
ste1=@res/@s;
graph date ste1;
?
? 係数ダミーと定数項ダミー
?
load d;
9*0 11*1;
dyd=d*yd;
print ce yd dyd d;
olsq ce c yd dyd d;
ste2=@res/@s;
graph date ste2;
end;
```